An Introduction to Maple

We will always work in what is called a **worksheet** .

In this environment **red** text after the Maple prompt (>) represents **Maple input** .

It can be executed by moving the cursor anywhere into that line and hitting the enter key (try this with any red line below).

Maple input usually is terminated by a semicolon (**;**).

The execution of a Maple command will usually be followed by some kind of **output** displayed in blue .

The display of the output (but not the actual calculation) can be suppressed by ending a command with a colon (**:**) instead of a semicolon (**;**).

Use the following to get acquainted with Maple's syntax and some of the commands needed more frequently in the assignments to come.

Other commands will be introduced when they are needed.

```
> restart;
```

This command is useful at the beginning of every worksheet. It clears Maple's memory.

Maple can do simple arithmetic. Here are a few examples:

```
> 2-3+4/5*6^7;
```

$$\frac{1119739}{5}$$

If you need to use floating point arithmetic then do

```
> evalf(1/3 + 1/2);
```

$$.8333333333$$

By default 10 digits are displayed, although this can be changed easily by adding the desired number of digits to the evalf command.

```
> evalf(1/3 + 1/2, 50);
```

$$.83333333333333333333333333333333333333333333333333$$

Maple also knows certain mathematical constants.

(Note the capitalization)

```
> Pi;
```

$$\pi$$

As a floating point number

> **evalf(Pi);**

$$3.141592654$$

Another constant that you might use is the number $e = 2.7182$ ...

This is how to get it in Maple:

> **exp(1);**

$$e$$

or

> **evalf(exp(1));**

$$2.718281828$$

A useful **shortcut** is the operator **%** . It refers to the most recent output.

For example (note again that Maple treats the following as an **exact** value)

> **sqrt(2);**

$$\sqrt{2}$$

> **%^3;**

$$2\sqrt{2}$$

does the same as

> **(sqrt(2))^3;**

$$2\sqrt{2}$$

Alternatively, we can use **assignments** .

In the previous example this would mean

> **root2 := sqrt(2);**

$$root2 := \sqrt{2}$$

Here we assigned the **value** sqrt(2)= $\sqrt{2}$ to the **variable** root2.

Thus, the following command will also produce the desired answer:

> **root2^3;**

$$2\sqrt{2}$$

The next Maple concept we need to understand is that of an **expression** .

All of the examples used above were expressions.

To illustrate the use of expressions better we make use of the real strength of Maple - it's **symbolic capabilities** .

We know that the area of a circle is given by the expression πr^2 .

Let's assign this expression to a variable **area_e** (the **_e** being used here to emphasize that we are dealing with an **e** xpression) :

> **area_e := Pi*r^2;**

$$area\_e := \pi r^2$$

So the expression Pi*r^2 is now stored under the name area_e.
If we want to know the area of a specific circle, say the one with r=3, we can do this as follows:

> **subs(r=3,area_e);**

$$9\pi$$

Note that we had to enter Pi*r^2 even though Maple displays the result as πr^2 .

If we omit the multiplication operator * - this is what will happen:

(You always have to include arithmetic operators when you are formulating Maple input.

This is a **common source of mistakes** !)

> **junk := Pir^2;**

$$junk := Pir^2$$

or

> **junk := Pi r^2;**

```
missing operator or `;`
```

As an alternative to using expressions (whose evaluation is sometimes a bit cumbersome) we can (and most often will) use **functions** .

Let's repeat the area example using function notation (now we use **area_f** for **f** unction).

Read this as "to every r the function area_f assigns the value πr^2 ":

```
> area_f := r -> Pi*r^2;
```

$$area\_f := r \to \pi r^2$$

Then the area of the circle with radius 3 is obtained by asking for

```
> area_f(3);
```

$$9 \pi$$

The difference between **expressions** and **functions** might seem confusing at the beginning.

Let's illustrate the difference in their use in connection with some other frequently used commands.

First we could plot the area as a function of the radius.

Using **expressions** we do

```
> plot(area_e, r=0..4);
```



In **function notation** the same is accomplished via

(note the missing r; Maple knows from the function definition above that r is the independent variable)

> **plot(area_f, 0..4);**



However, the following also works

> **plot(area_f(r), r=0..4);**



This shows us that we can easily get an **expression** from a **function** :

> **area_f(r);**

$$\pi r^2$$

is the same as

> **area_e;**

$$\pi r^2$$

How do we go the other way, i.e., how can we turn an expression into a function?

This is done with the help of the command **unapply** .

Here is how it works (for a similar example):

> **circumference := 2*Pi*r;**

$$circumference := 2\,\pi\,r$$

So circumference holds the **expression** 2*Pi*r.

To get a **function** we do

(i.e., we tell Maple which expression to convert to a function, and what the independent variable will be):

> **circumf := unapply(circumference, r);**

$$circumf := r \rightarrow 2\,\pi\,r$$

We end this introduction with a simple calculus problem.

First we will define a simple function

> **f := x -> sin(x/2)*(x-1);**

$$f := x \rightarrow \sin\left(\frac{1}{2}x\right)(x-1)$$

Let's plot the graph of f on the interval $[-2\,\pi, 2\,\pi]$.

> **plot(f, -2*Pi..2*Pi);**



To find the intersections of the graph with the x -axis we can attempt to use either **solve** or **fsolve**

(note that both functions cannot be blindly trusted; they often find only some - but not all solutions)

> **solve(f(x)=0, x);**

We missed the intersections at the endpoints of the interval.

$$0, 1$$

Using fsolve on the entire interval even gives only one solution

> **fsolve(f(x)=0, x, -2*Pi..2*Pi);**

$$0$$

However, some fine tuning helps

> **fsolve(f(x)=0, x, 1/2..2);**

$$1.$$

and (e.g.)

> **fsolve(f(x)=0, x, -6.5..-6);**

$$-6.283185307$$

To locate critical points we need to know the **derivative** of f .

Since we are using **function notation** this is done via

> **D(f);**

$$x \to \frac{1}{2}\cos\left(\frac{1}{2}x\right)(x-1)+\sin\left(\frac{1}{2}x\right)$$

Expressions are differentiated with the help of **diff**

> **diff(f(x), x);**

$$\frac{1}{2}\cos\left(\frac{1}{2}x\right)(x-1)+\sin\left(\frac{1}{2}x\right)$$

We can plot the graph of the derivative on the same interval as above

> **plot(D(f), -2*Pi..2*Pi);**



Or even both graphs together

> **plot({f, D(f)}, -2*Pi..2*Pi);**



Finally, we can compute integrals.

The antiderivative of the derivative of f should be f again

(note that Maple doesn't bother with the additive constant):

> **int(D(f)(x), x);**

$$x\sin\left(\frac{1}{2}x\right)-\sin\left(\frac{1}{2}x\right)$$

Note that we integrated an **expression** above, and Maple also returned an expression.

It does not seem to be possible to antidifferentiate a **function** in Maple

(or obtain the result of integration as a function - if you want this you need to use **unapply**).

Also note that the answer does not look quite like what we expected.

Another useful command is **simplify** . It might help in situations like this.

> **simplify(%);**

$$x\sin\left(\frac{1}{2}x\right)-\sin\left(\frac{1}{2}x\right)$$

It doesn't here.

Since we really started with a factored version of this answer let's try

> **factor(%);**

$$\sin\left(\frac{1}{2}x\right)(x-1)$$

To compute a **definite integral** we simply add the limits of integration as in

> **int(f(x), x=-2*Pi..2*Pi);**

$$8\pi$$

As a simple exercise you might want to compute the zeros of the derivative of f above:

>